

## I. PRZEBIEG ĆWICZENIA

A. Programy podane w ćwiczeniu LV uruchom na drugiej linijce diodowej (zielonej), w taki sposób, aby na obu linijkach wyświetlało się dokładnie to samo (a nie stanowiło negacji).

Wskazówki:

- zwróć uwagę na odmienny sposób sterowania diodami w linijce zielonej i czerwonej (poziom niski na porcie P2 zapala diody zielone, poziom wysoki na porcie P2 zapala diody czerwone),
- nie mogą być aktywne jednocześnie dwie linijki diodowe (tylko jedna linia P3.3 lub P3.5 może być w danym momencie aktywna/ustawiona w stan niski),
- odstęp pomiędzy kolejnymi przerwaniem, zmieniającymi aktywną linijkę diodową, powinien być mniejszy od 10 ms (większy odstęp może spowodować widoczne miganie linijek diodowych).

### PRZYKŁAD

NAME sterowanie\_dynamiczne

T0\_r EQU 0CCCCCh

IRAM\_r EQU 07Fh

LED\_cz EQU P3.3

LED\_z EQU P3.5

-----  
BSEG at 0h ;segment w obszarze bitowym BDATA

FLAGA DBIT 01h ;zarezerwuj bit zerowy bajtu 20h,  
; (zmienna bitowa o nazwie Flaga)

-----  
DSEG at 21h ;segment danych w obszarze DATA

L\_cz DS 01h ;deklaracja jednobajtowej zmiennej (adres 21h w DATA)

L\_z DS 01h ;dyrektywa DS rezerwuje na zmienną podaną liczbę bajtów

-----

CSEG at 0000h  
ljmp Start

-----  
ORG 000Bh  
ljmp Int\_T0

-----

Start:

```

mov SP,#60 ;wskaźnik stosu domyślnie jest ustawiony na 07h
orl TMOD,#0000$0001b ;inicjalizacja układu czasowo - licznikowego
mov TH0,#HIGH(T0_r)
mov TL0,#LOW(T0_r)
orl IE,#1000$0010b
setb TR0

```

```

mov R0,#IRAM_r ;czyszczenie pamięci wewnętrznej
clr A

```

Reset:

```

mov @R0,A
djnz R0,Reset

```

```

Dodaj:
    mov R1,#0000$0001b    ;zapalenie kolejnej diody
Dodaj1:
    mov A,L_cz
    orl A,R1
    mov L_cz,A           ;ustawienie wartości wyświetlanej na LED czerw.
    xrl A,#1111$1111b
    mov L_z,A           ;ustawienie wartości wyświetlanej na LED zielonych
    mov A,R1
    r1 A
    mov R1,A
    cjne R0,#100,$      ;opóźnienie zmieniania wyświetlanych wartości
    mov R0,#0
    cjne R1,#0000$0001b,Dodaj1 ;przejsście do kolejnego cyklu - zmiana pętli

Odejmiuj:
    mov R1,#1111$1110b    ;gaszenie kolejnej diody
Odejmiuj1:
    mov A,L_cz
    anl A,R1
    mov L_cz,A
    xrl A,#1111$1111b
    mov L_z,A
    mov A,R1
    r1 A
    mov R1,A
    cjne R0,#100,$
    mov R0,#0
    cjne R1,#1111$1110b,Odejmiuj1
    ljmp Dodaj
;-----
Int_T0:    ;obsługa przerwania od T0
    mov TH0,#HIGH(T0_r)
    mov TL0,#LOW(T0_r)
    inc R0
    cpl FLAGA           ;FLAGA znacznik realizowanej części podprogramu
    jb FLAGA,Int_cz    ;zmiana aktywnej linijki LED
    setb LED_cz         ;wyświetlanie informacji na diodach zielonych
    mov P2,L_z          ;pobieranie informacji ze zmiennej L_z
    clr LED_z
    reti
Int_cz:
    setb LED_z          ;wyświetlanie informacji na diodach czerwonych
    mov P2,L_cz         ;pobieranie informacji ze zmiennej L_cz
    clr LED_cz
    reti
;-----
END

```

B. Napisz program uruchamiający na linijce czerwonej inny niż na linijce zielonej podprogram z ćwiczenia LV.

### **PRZYKŁAD**

```

NAME dwa_programy
T0_r      EQU    0CCCCb
IRAM_r    EQU    07Fh
LED_cz    EQU    P3.3
LED_z     EQU    P3.5

```

```

BSEG at 0h
F_cz_z          DBIT 01h
F_czas          DBIT 01h

DSEG at 21h
L_cz           DS   01h
L_z           DS   01h
;-----
CSEG at 0000h
    ljmp Start
;-----
ORG 000Bh
    ljmp Int_T0
;-----
Start:
    mov SP,#60
    orl TMOD,#0000$0001b
    mov TH0,#HIGH(T0_r)
    mov TL0,#LOW(T0_r)
    orl IE,#1000$0010b
    setb TR0

    mov R0,#IRAM_r
    clr A
Reset:
    mov @R0,A
    djnz R0,Reset

    mov R1,#0000$0001b
    mov L_z,#1111$1111b
Czekaj:
    jnb F_czas,Czekaj
    clr F_czas

    mov A,L_cz                ;zmiana diod czerwonych
    cjne A,#1111$1111b,Czerwona
    clr A
    mov L_cz,A
    jmp Zielona
Czerwona:
    orl A,R1
    mov L_cz,A
    mov A,R1
    rl A
    mov R1,A
Zielona:                ;zmiana diod zielonych
    mov A,L_z
    cpl A
    inc A
    cpl A
    mov L_z,A
    ljmp Czekaj
;-----
Int_T0:
    mov TH0,#HIGH(T0_r)
    mov TL0,#LOW(T0_r)
    inc R0
    cjne R0,#100,Int_licz
    mov R0,#0
    setb F_czas

```

```

Int_licz:
    cpl F_cz_z
    jb F_cz_z,Int_cz
    setb LED_cz
    mov P2,L_z
    clr LED_z
    reti
Int_cz:
    setb LED_z
    mov P2,L_cz
    clr LED_cz
    reti
;-----
END

```

## II. PRZYGOTOWANIE DO NASTĘPNYCH ZAJĘĆ

### 1. Wiedza teoretyczna:

A. Zasada dynamicznej obsługi wyświetlaczy.

### 2. Wiadomości z ćwiczenia siódmego:

A. Znajomość sposobu dynamicznej obsługi linijek diodowych,

B. Znajomość kodu źródłowego pisanych programów.

### 3. Zastępcza forma zaliczenia wiadomości z ćwiczenia siódmego.

**Ręcznie** na papierze kancelaryjnym napisać kody źródłowe programów podanych w tym ćwiczeniu wraz wyjaśnieniem **przeznaczenia** (w jakim celu użyto dany rozkaz?) każdej linijki programu.